



Embedded Systems Design (630470)

Lecture 12

Keypad Interfacing

Prof. Kasim M. Al-Aubidy


Philadelphia University

- Human interfacing needs and some simple means of meeting these.
- Some simple example sensors
- Some ways of interfacing between sensor signals and the microcontroller
- Some simple example actuators
- Some ways of interfacing between the microcontroller and the actuator
- Embedded Systems Design: Projects


- **The main idea – the human interface:**
- The human has to interface with any machine that he/she works with. This is almost inevitably in some form of closed loop interaction.




Hand Controller:



(a). LED Version




(b). LCD Version.




Keypad Interfacing:


- A keypad is based on switches, each switch is connected in a matrix.



Keypad circuit diagram, with pull-up resistors



Key	Value Read
1	0111 011X
2	0111 101X
3	0111 110X
4	1011 011X
5	1011 101X
6	1011 110X
7	1101 011X
8	1101 101X
9	1101 110X
*	1110 011X
0	1110 101X
#	1110 110X




```

;*****
;keypad_test
;Tests keypad, writing key pressed to lcd display
;*****
...
(opening program sections omitted)
...
;Initialise
    bsf    status,rp0    ;select memory bank 1
...
    movlw B'11110000'   ;Port B initially Row bits ip, column op
    movwf trisb         ;(port B not used)
    bcf    status,rp0   ;select bank 0
...
(lcd initialisation omitted)
...
    clrf  portb ;initialise keypad value
;enable interrupt
    bcf  intcon,rbif
    bsf  intcon,rbie
    bsf  intcon,gie
loop  goto  loop        ;await keypad entries

```

```

;*****
;Interrupt Service Routine.
;*****
;Keypad press has been detected through Port B Interrupt on Change.Gets
;keypad pattern, converts to character, stores in kpad_char, sends to lcd,
;and awaits key release,
kpad_to_lcd call    kpad_rd
;now convert code to character, forming address used in lookup table
        call  kp_code_conv
;now send to lcd
        bsf   portc,lcd_RS ;set for character op
        movwf lcd_op
        call  lcd_write
;test now for keypad release
rel_test call kpad_rd
        movf  kpad_pat,0
        andlw 0fe          ;suppress lsb, which is not used
        sublw 0fe          ;test if inactive
        btfss status,z
        goto  rel_test
        bcf   intcon,rbif  ;clear interrupt flag
        retfie
;
;*****

```

```

;*****
;SUBROUTINES
;*****
;Reads keypad, places pattern into kpad_pat, and resets keypad interface
kpad_rd movf portb,w      ;read portb value, this will be row pattern
        andlw B'11110000' ;ensure unwanted bits are suppressed
        movwf kpad_pat
        bsf  status,rp0   ;set row to op, column to ip
        movlw B'00001110'
        movwf trisb
        bcf  status,rp0
        movlw 00
        movwf portb      ;ensure output values still zero
        movf  portb,w    ;read portb value, this will be column pattern
        andlw B'00001110' ;ensure unwanted bits are suppressed
        iorwf kpad_pat,1 ;OR those results into the pattern
;reset keypad interface
        bsf  status,rp0 ;set row to ip, column to op
        movlw B'11110000'
        movwf trisb
        bcf  status,rp0
        clrf portb ;ensure output values still zero
        return

```

;Converts keypad pattern held in kpad_pat to ASCII character, first forming
;address (in kpad_add) that is used in lu table. Returns with character held
;in kpad_char

```
kp_code_conv bcf status,c
             rrf  kpad_pat,1    ;discard bit 0 which is not used
             clrf kpad_add
;deduce row
             btfsc kpad_pat,6
             goto kp1
             goto col_find      ;here if row 1, kpad_add stays as is
kp1         btfsc kpad_pat,5
             goto kp2
             movlw B'00000100'  ;here if row 2
             iorwf kpad_add,1    ;form table address
             goto col_find
kp2         btfsc kpad_pat,4
             goto kp3
             movlw B'00001000'  ;here if row 3
             iorwf kpad_add,1    ;form table address
             goto col_find
kp3         btfsc kpad_pat,3
             goto kp4
             movlw B'00001100'  ;here if row 3
             iorwf kpad_add,1    ;form table address
             goto col_find
```

```

kp4    movlw D'16'           ;no row detected, return "E" via Table
      goto keypad_op
;now deduce column
col_find btfsc      kpad_pat,2
      goto  cf1
      goto  keypad_op      ;here if column 1, kpad_add stays as is
cf1    btfsc kpad_pat,1
      goto  cf2
      movlw B'00000001'    ;here if column 2
      iorwf kpad_add,1     ;form table address
      goto  keypad_op
;assume now column 3
cf2    movlw B'00000010'
      iorwf kpad_add,1     ;form table address
keypad_op movf kpad_add,0
      call  kp_table
      movwf kpad_char      ;save the character
      return

```

;Table called to convert pattern recd from keypad to actual character. Note that
;ASCII codes will be returned, as each digit is in format 'D'.

```
kp_table addwf pcl,1
    retlw '1'           ;row 1
    retlw '2'
    retlw '3'
    retlw 'A'          ;Error code
    retlw '4'          ;row 2
    retlw '5'
    retlw '6'
    retlw 'B'          ;Error code
    retlw '7'          ;row 3
    retlw '8'
    retlw '9'
    retlw 'C'          ;Error code
    retlw '*'          ;row 4
    retlw '0'
    retlw '#'
    retlw 'D'          ;Error code
    retlw 'E'          ;Error code
```

